

# Příkazový jazyk pro vytváření předpisů telefonického připojení

Copyright (c) 1997 Microsoft Corp.  
Část tohoto materiálu byla dodána společností Shiva Corporation.

## Použití tohoto dokumentu

Tento dokument je určen především zprostředkovatelům přístupu k síti Internet (ISP) k vytváření předpisů, které mohou koncoví uživatelé upravovat pro svá připojení.

## Použití předpisů v aplikaci Internet Explorer

Pomocí programu IEScript.exe může koncový uživatel použít k připojení soubor s předpisem (.scp). Jak je dále popsáno, IESCRIPIT přidá nastavení k souboru připojení (.con). K úpravě souboru připojení použijte IESCRIPIT—připojení neupravujte ručně.

Název sekce: **Script**

Je možné použít následující klíčová slova:

### **ScriptEnabled**

Yes/No. Určuje, zda se má aktivovat program pro předpisy.

### **ScriptFileName**

Úplný řetězec určující umístění souboru s předpisem, který byl přiřazen tomuto souboru připojení.

### **ScriptTerminal**

Yes/No. Určuje, zda se má během přehrávání předpisu zavést okno terminálu s předpisem.

### **ScriptRecord**

Yes/No. Určuje, zda má při aktivaci spustitelný předpis přejít do režimu nahrávání.

### **Ukázky částí souboru připojení pro vytváření předpisů**

#### **Příklad 1:**

```
[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Myscript.Scp
ScriptTerminal=Yes
ScriptRecord=No
```

#### **Příklad 2:**

```
[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\NewScrip.Scp
ScriptTerminal=Yes
ScriptRecord=Yes
```

### **Poznámka**

- Příklad 2 ukazuje zobrazení v průběhu nahrávání předpisu.

## Vytvoření předpisu v aplikaci Internet Explorer

Vytváříte-li soubor s předpisem pro koncové uživatele, je třeba nejprve nahrát předpis pomocí

programu Scriptor.exe. Po vytvoření tohoto souboru může koncový uživatel soubor upravit a vložit vlastní jméno a heslo.

### Nahrání a přehrání předpisu pomocí okna TTY

- 1 Spustíte program Scriptor.exe. Obvykle je uložen ve stejném adresáři jako aplikace Internet Explorer.
- 2 V nabídce Soubor klepněte na příkaz Nahrát.
- 3 Vyhledejte soubor s předpisem (.scp) a vyberte jej ze seznamu souborů. Můžete také zadat nový název souboru.
- 4 Otevřete skupinu programů Internet Explorer a vyberte položku Nové připojení. Podle pokynů na obrazovce vytvořte nové připojení (nebo vyberte ikonu připojení, které již bylo vytvořeno).
- 5 Vyberte ikonu připojení. V nabídce Soubor klepněte na příkaz Vlastnosti. Ověřte, zda je zaškrtnuto políčko Okno terminálu zobrazit po vytvoření čísla.
- 6 Dokončete připojení k serveru sítě Internet.  
V okně programu ShivaRemote se zobrazí výzva "Started!!!".
- 7 Klepněte na okno záznamu (které teď pracuje jako okno terminálu) a potom zadejte informace potřebné pro přihlášení k serveru zprostředkovatele ISP.
- 8 Po přihlášení klepněte na příkaz Konec nahrávání v okně Zpracovat předpis.
- 9 V okně Zpracování předpisu od ShivaPPP klepněte na tlačítko OK.
- 10 Komunikaci ukončete klepnutím na tlačítko Zrušit v dialogovém okně připojení.

#### Poznámka

Uživatel si při používání tohoto předpisu musí ověřit, že není zaškrtnuto políčko Okno terminálu zobrazit po vytvoření čísla.

Po ukončení zadávání požadovaných informací do předpisu klepněte na tlačítko Stop. Okno se zavře a začne normální přenos po protokolu PPP.

Předpis vytvořený programem vypadá takto:

```
; C:\SCRIPTS\IE\LOGIN.SCP
; Created: 7-1-1996 at 17:07:11
;
;
proc main
string szPassword
transmit "^M"
waitfor "Host Name: ",matchcase until 15
transmit "spry01^M"
waitfor "UIC: ",matchcase until 6
transmit "spry053514^M"
waitfor "Password: ",matchcase until 3
if $PASSWORD then
transmit $PASSWORD
transmit "^M"
goto doneTxPassword
endif
getinput "Password: " szPassword
transmit szPassword
transmit "^M"
doneTxPassword:
endproc
```

### Úprava souboru s předpisem založeném na nahraném souboru nebo vytvoření nového souboru s předpisem

Program Scriptor.exe nahraje soubor s předpisem, který obsahuje informace týkající se uživatele včetně jeho jména a hesla. Zprostředkovatel připojení k síti Internet může poskytovat obecný soubor s předpisem, který pak může libovolný koncový uživatel upravit a používat.

Jako návod pro vytvoření obecného souboru s předpisem který neobsahuje konkrétní informace o uživateli, může sloužit následující ukázka předpisu.

```
;-----  
; Název souboru: LOGIN01.SCP  
; Datum vytvoření: 10-18-1996  
; Čas vytvoření: 12:00:00  
;-----  
  
proc main  
  
; V některých systémech je třeba iniciovat přihlášení  
;stisknutím klávesy.  
;-----  
  
transmit "^M"  
  
; Odeslání jména hostitele pevně zadaného  
; zprostředkovatelem služeb  
;-----  
  
waitfor "Host Name:"  
transmit "hostname"  
transmit "^M"  
  
; Odeslání jména uživatele převzatého z pole  
; Jméno uživatele v přiřazeném souboru s připojením  
;-----  
  
waitfor "Username:"  
transmit $USERID  
transmit "^M"  
  
; Odeslání hesla převzatého z pole  
; Heslo přiřazeného souboru s připojením  
;-----  
  
waitfor "Password:"  
transmit $PASSWORD  
transmit "^M"  
  
endproc
```

## Podrobnější informace o předpisech

### 1.0 Přehled

Mnoho zprostředkovatelů připojení k síti Internet a informačních služeb požaduje, aby uživatel ručně zadal některé informace, např. jméno a heslo, nutné pro připojení. Předpisy pro telefonické připojení umožňují uživateli tento proces automatizovat.

Předpis je textový soubor obsahující posloupnost příkazů, parametrů a výrazů, které vyžaduje zprostředkovatel připojení k síti Internet nebo informačních služeb, aby mohl navázat připojení

a zprostředkovat využití služby. K vytvoření souboru s předpisem lze použít libovolný textový editor, například Poznámkový blok společnosti Microsoft. Po vytvoření souboru s předpisem lze tento soubor přiřadit konkrétnímu připojení pomocí nástroje Předpisy pro telefonické připojení.

## 2.0 Základní struktura předpisu

Základní instrukce, kterou soubor s předpisem obsahuje, je příkaz. Některé příkazy vyžadují parametry, které dále definují funkci příkazu. Výraz je kombinace operátorů a argumentů; vyhodnocením výrazu vznikne výsledek. V každém příkazu lze výrazy používat jako hodnoty. Příkladem výrazů jsou aritmetické výrazy, relační porovnávání nebo spojení řetězců.

Následuje příklad základní formy předpisu pro telefonické připojení:

```
;
; Komentář začíná středníkem a pokračuje
; do konce řádku.
;

proc main
    ; Předpis může obsahovat libovolný počet
    ; proměnných a příkazů.

    deklarace proměnných

    příkazový blok

endproc
```

Předpis musí obsahovat hlavní proceduru uvedenou klíčovým slovem **proc** a odpovídajícím klíčovým slovem **endproc** označený konec procedury.

Proměnné je třeba deklarovat před přidáním příkazů. Nejdříve se spustí první příkaz v hlavní proceduře a po něm následují ostatní příkazy v pořadí, v jakém jsou uvedeny v předpisu. Předpis skončí po dosažení konce hlavní procedury.

## 3.0 Proměnné

Předpisy mohou obsahovat proměnné. Názvy proměnných musí začínat písmenem nebo podtržítkem ( \_ ) a mohou obsahovat libovolnou posloupnost malých a velkých písmen, číslic a podtržení. Jako název proměnné nelze použít vyhrazená slova. Další informace viz seznam vyhrazených slov na konci dokumentu.

Proměnné je třeba deklarovat před jejich použitím. Při deklaraci proměnné musí být také definován její typ. Proměnná určitého typu může nabývat jen hodnot stejného typu. Je možné používat následující tři typy proměnných:

<u>Typ</u>	Integer
<u>Popis</u>	Záporné nebo kladné číslo, například 7, -12 nebo 5698.
<u>Typ</u>	String
<u>Popis</u>	Posloupnost znaků uzavřená do uvozovek—například "Nazdar světe!" nebo "Zadejte heslo:".
<u>Typ</u>	Boolean
<u>Popis</u>	Logická hodnota TRUE nebo FALSE.

K přiřazení hodnot proměnným se používá následující přiřazovací příkaz:

```
proměnná = výraz
```

Proměnné je přiřazena vyhodnocená hodnota.

Příklady:

```

integer count = 5
integer timeout = (4 * 3)
integer i

boolean bDone = FALSE

string szIP = (getip 2)

set ipaddr szIP

```

Výše uvedený příklad obsahuje předpisové příkazy které systém Shiva nepodporuje.

### 3.1 Systémové proměnné

Systémové proměnné se nastavují příkazy předpisu nebo jsou určeny informacemi zadanými při nastavení připojení. Systémové proměnné slouží pouze pro čtení, což znamená, že je v rámci předpisu nelze měnit. Existují tyto systémové proměnné:

<u>Název</u>	\$USERID
<u>Typ</u>	String
<u>Popis</u>	Identifikace uživatele pro aktuální připojení. Tato proměnná má hodnotu jména uživatele zadaného v dialogovém okně Připojení programu Připojení.

<u>Název</u>	\$PASSWORD_
<u>Typ</u>	String
<u>Popis</u>	Heslo pro aktuální připojení. Tato proměnná má hodnotu jména uživatele zadaného v dialogovém okně Připojení programu Připojení.

<u>Název</u>	\$SUCCESS
<u>Typ</u>	Boolean
<u>Popis</u>	Tato proměnná se nastavuje určitými příkazy jako indikace úspěšnosti průběhu příkazu. Na základě hodnoty této proměnné může předpis provádět rozhodnutí.

<u>Název</u>	\$FAILURE
<u>Typ</u>	Boolean
<u>Popis</u>	Tato proměnná se nastavuje určitými příkazy jako indikace neúspěšnosti průběhu příkazu. Na základě hodnoty této proměnné může předpis provádět rozhodnutí.

Tyto proměnné lze používat na všech místech, kde se používá výraz obdobného typu. Například

```
transmit $USERID
```

je platný příkaz, protože \$USERID je proměnná typu string.

### 4.0 Řetězcové literály

Zpracování předpisů v nástroji Předpisy pro telefonické připojení umožňuje používání následujících escape sekvencí a přenosů se stříškou:

<u>Řetězcový literál</u>	<u>Popis</u>
<u>Řetězcový literál</u>	<u>^znak</u>
<u>Popis</u>	přenos se stříškou

Je-li *znak* hodnota mezi '@' a '\_', přeloží se znaková posloupnost na jednobajtovou hodnotu od 0 do 31. Například ^M se převede na znak zalamování řádku.

Je-li *znak* hodnota mezi 'a' and 'z', přeloží se znaková posloupnost na jednobajtovou hodnotu od 1 do 26.

Má-li *znak* jinou hodnotu, znaková posloupnost se nezpracuje.

<u>Řetězcový literál</u> <u>Popis</u>	<cr>	Zalamování řádku
<u>Řetězcový literál</u> <u>Popis</u>	<lf>	Posuv o řádek
<u>Řetězcový literál</u> <u>Popis</u>	\"	Uvozovka
<u>Řetězcový literál</u> <u>Popis</u>	\^	Stříška
<u>Řetězcový literál</u> <u>Popis</u>	\<	Znak 'menší než' (<)
<u>Řetězcový literál</u> <u>Popis</u>	\\	Zpětné lomítko

Příklady:

```
transmit "^M"  
transmit "Joe^M"  
transmit "<cr><lf>"  
waitfor "<cr><lf>"
```

## 5.0 Výrazy

Výraz je kombinace operátorů a argumentů, jejichž vyhodnocením vznikne výsledek. Ve všech příkazech lze výrazy používat jako hodnoty.

Ve výrazu mohou být kombinovány proměnné nebo hodnoty typu integer, string nebo boolean spolu s unárními a binárními operátory z následující tabulky. Všechny unární operátory mají nejvyšší prioritu. Priorita binárních operátorů odpovídá jejich pozici v tabulce.

Užívají se následující unární operátory:

<u>Operátor</u> <u>Typ operace</u>	-	Unární minus
<u>Operátor</u> <u>Typ operace</u>	!	Jedničkový doplněk

Binární operátory jsou uvedeny v následující tabulce v pořadí podle priority. Operátory s vyšší prioritou jsou uvedeny nejdříve:

<u>Operátory</u> <u>Typ operace</u> <u>Typová vymezení</u>	* /	Násobení Integer
<u>Operátory</u> <u>Typ operace</u> <u>Typová vymezení</u>	+ -	Přidávání pro typ integer String (pouze kladné)

<u>Operátory</u>	< > <= >=
<u>Typ operace</u>	Relační
<u>Typová vymezení</u>	Integer
<u>Operátory</u>	== !=
<u>Typ operace</u>	Rovnost
<u>Typová vymezení</u>	Integer, string, boolean
<u>Operátory</u>	and
<u>Typ operace</u>	Logický součin
<u>Typová vymezení</u>	Boolean
<u>Operátory</u>	or
<u>Typ operace</u>	Logický součet
<u>Typová vymezení</u>	Boolean

Příklady:

```
count = 3 + 5 * 40
transmit "Dobrý" + " den"
delay 24 / (7 - 1)
```

## 6.0 Komentáře

Veškerý text na řádku, který začíná středníkem, se ignoruje.

Příklady:

```
; toto je komentář

transmit "ahoj" ; přenese řetězec "ahoj"
```

## 7.0 Klíčová slova

Klíčová slova určují strukturu předpisu. Na rozdíl od příkazů neprovádějí žádné akce. Seznam klíčových slov je uveden dále.

### **proc** *název*

Označuje začátek procedury. Ve všech předpisech musí být hlavní procedura (**proc main**). Provádění předpisu začíná hlavní procedurou a končí na konci hlavní procedury.

### **endproc**

Označuje konec procedury. Když předpis dojde na příkaz **endproc** hlavní procedury, nástroj Předpisy pro telefonické připojení spustí protokoly PPP nebo SLIP.

### **integer** *název* [ = *hodnota* ]

Deklaruje proměnnou typu integer. K inicializaci proměnné se používá libovolný numerický výraz nebo proměnná.

### **string** *název* [ = *hodnota* ]

Deklaruje proměnnou typu string. K inicializaci proměnné se používá libovolný řetězcový literál nebo proměnná.

### **boolean** *název* [ = *hodnota* ]

Deklaruje proměnnou typu boolean. K inicializaci proměnné se používá libovolný logický výraz nebo proměnná.

## 8.0 Příkazy

Výchozí implementace předpisového jazyka pro vytáčení telefonních čísel aplikace Internet Explorer je podmnožinou příkazového jazyka pro vytváření předpisů telefonického připojení, který je předepsán pro systém Microsoft Windows® 95.

Všechny příkazy jsou vyhrazená slova, nelze tedy deklarovat proměnné se stejnými názvy, jako mají příkazy. Používají se následující příkazy:

### **delay** *nSekund*

Před provedením dalšího příkazu v předpisu počká počet sekund daný hodnotou *nSekund*.

Příklady:

```
delay 2          ; počká 2 sekundy
delay x * 3      ; počká x * 3 sekundy
```

### **goto** *návěští*

Přejde na místo v předpisu dané návěští a pokračuje zpracováním příkazů po něm následujících.

Příklad:

```
waitfor "Prompt>" until 10
if !$SUCCESS then
    goto BailOut ; skočí na BailOut a provede příkazy
                    ; za ním
endif

transmit "bbs^M"
goto End

BailOut:
transmit "^M"
```

### **halt**

Zastaví předpis. Příkaz neodstraní dialogové okno terminálu. K navázání spojení je třeba klepnout na tlačítko Pokračovat. Předpis nelze restartovat.

### **if** *podmínka* **then** *příkazy* **endif**

Má-li *podmínka* hodnotu TRUE, provede posloupnost *příkazů*.

Příklad:

```
if $USERID == "John" then
    transmit "Johnny^M"
endif
```

### *návěští* :

Určuje místo v předpisu, na které se má přejít. Návěští musí být jednoznačené a odpovídat konvenci pro pojmenování proměnných.



## **transmit řetězec [ , raw ]**

Odešle znaky dané *řetězcem* na vzdálený počítač.

Není-li v příkazu obsažen parametr **raw**, rozpozná vzdálený počítač sekvence escape a přenosy se stříškou. Parametr **raw** se používá při přenosu systémových proměnných \$USERID a \$PASSWORD, kdy jméno uživatele a heslo obsahují znakové posloupnosti, které by bez parametru **raw** byly interpretovány jako sekvence escape nebo přenosy se stříškou.

Příklady:

```
transmit "slip" + "^M"  
transmit $USERID, raw
```

## **waitfor řetězec [ , matchcase ] [ then návěští { , řetězec [ , matchcase ] then návěští } ] [ until doba ]**

Čeká, dokud počítač nepřijme od vzdáleného počítače jeden nebo několik zadaných řetězců. Pokud nezadáte parametr **matchcase**, parametr *řetězec* nerozlišuje malá a velká písmena.

Je-li přijat odpovídající řetězec a pak použit parametr **then návěští**, přejde tento příkaz v předpisu na místo určené *návěštím*.

Volitelný parametr **until doba** definuje maximální počet sekund, po které počítač čeká na přijetí řetězce před provedením dalšího příkazu. Bez tohoto parametru čeká počítač po neomezeně dlouhou dobu.

Přijme-li počítač jeden ze zadaných řetězců, nastaví se systémová proměnná \$SUCCESS na hodnotu TRUE. V případě, že stanovená čekací doba uplyne před přijetím řetězce, proměnná se nastaví na hodnotu FALSE.

Příklady:

```
waitfor "Login:"  
  
waitfor "Password?", matchcase  
  
waitfor "prompt>" until 10  
  
waitfor  
    "Login:"      then DoLogin,  
    "Password:"  then DoPassword,  
    "BBS:"       then DoBBS,  
    "Other:"     then DoOther  
until 10
```

Následující příkaz je rozšířením sady příkazů pro tvorbu předpisů společnosti Microsoft dodaným společností Shiva.

## **getinput "Zobrazit řetězec" szToReceiveInput**

Vyzve uživatele k zadání informací v průběhu provádění předpisu.

Příklad:

```
Getinput "System Password" szSystemPassword
```

## **9.0 Vyhrazená slova**

Následující slova jsou vyhrazena a nelze je používat jako názvy proměnných:

and	boolean	databits	delay
do	endif	endproc	endwhile
even	FALSE	getip	goto
halt	if	integer	ipaddr
keyboard	mark	matchcase	none
odd	off	on	or
parity	port	proc	raw
screen	set	space	stopbits
string	then	transmit	TRUE
until	waitfor	while	

## 10.0 Příkazová slova systému Shiva, která Zpracování předpisu nepodporuje

- getip
- port databits
- port parity
- port stopbits
- set screen keyboard
- ipaddr
- while/endwhile